
In our first paper we presented a BASIC computer program for performing the McCabe-

Thiele (1) theoretical stage step-off procedure for a single feed – two product binary 

distillation column.  We emphasized application to close boiling separations such as for 

splitter fractionators where the operating and equilibrium lines (curves) are generally 

quite close together and where nearly constant molal overflow prevails.   

In this second installment we extend the computer program to handle a two feed – two 

product distillation column and also incorporate the capability to calculate theoretical 

tray temperatures.  In principle this procedure could be applied to any number of feeds 

where a large number of intermediate operating lines are involved. 

Working relationships.  First of all, the working relationships used in developing the 

program will be presented.  Our focus here will be restricted to two feed – two (liquid) 

product distillations.  Figure 1 describes the column configuration and nomenclature 

employed.  Here we have three sections – rectifying, middle and stripping. 

Equilibrium curve.  Once again the binary vapor-liquid coexistence curve is represented 

by Eqs. 1 and 2. 

where x and y are the mole fractions of the light or lower boiling component in the liquid 

and vapor respectively.   is the relative volatility of the light component with respect to 

the heavier component (Klt/Khvy).  The constants A, B and C of Eq. 1 are fit to a set of 

isobaric binary VLE data, which hopefully cover the full range of composition for the 

binary mixture to be distilled.  If VLE data are not available, then  values for the fit will 

have to be generated by some other means such as from a generalized correlation which 

is believed to give reasonably accurate predictions.  Tray-by-tray temperatures can also 

be calculated if we also have the bubble point (T-x) coexistence curve as well.  This 

curve can be well represented by a similar equation form as used for the relative 

volatility, namely, 

2 (3)T E x F x G    

Since Eqs. 1 and 3 are parabolic in nature, we must be sure that neither a maximum or 

minimum (extrema) occurs in either one over the range 0  x  1. 

The q-lines.  For a two-feed distillation column there will be two q-lines: 

 In either equation q is defined as the moles of liquid introduced to the section 

immediately below the feed tray in question resulting from each mole of feed.  It 

basically characterizes the phase condition (mole fraction liquid) of the feed as it enters 

the column ie. 

Below is listed a summary of the numerical limits for q depending upon the various phase 

conditions of the entering feed: 

  Phase Condition   q 

  Cold feed    > 1 

  Bubble pt. feed   = 1 

  Two phase feed           0 < q < 1   

  Dew pt. feed    = 0    

  Superheated vapor feed  < 0 

Minimum reflux ratio.  The minimum reflux ratio (L/D)min is determined in the same 

fashion as for a single feed column.  It is calculated by finding the point where the top q-

line intersects the equilibrium curve.  This is Point a shown on Figure 2.  Eqs. 1,2 and 4 

are combined to eliminate y, and the resulting expression is solved by trial and error 

substitution to give x
/
, the liquid mole fraction at Point a.  y

/
 is then calculated from Eq. 2 

or 4.   The minimum reflux ratio is then given by, 

/

/ /
(8)D

M

M

x yL
R

D y x

 
  

 
    

 

 



 

-4- 

Upper operating line.  The equation of the upper operating line (UOL) for constant 

molal overflow is  

(9)

It is represented as line segment b-c on Figure 2.  Here R is the actual operating reflux 

ratio (L/D) and is established by some specified percentage over the minimum, RM.  In 

order to determine the intersection point of the UOL and the upper q-line (Point b), it is 

necessary to equate Eqs. 4 and 9.   The resulting expression is solved directly for x
//
, the 

liquid composition at this point of intersection.  y
//
 is then determined by either Eq. 4 or 9. 

Middle operating line.  Before the middle operating line (MOL) can be established, an 

overall material balance must be performed in order to establish the overhead and 

bottoms product rates.  The result is, 

Here both the upper and lower feed rates and their compositions (xF1, xF2) must be 

specified. 

Next the liquid and vapor flow rates in the rectifier (upper section) are determined by, 

The liquid and vapor flow rates in the middle section are established by using the 

definition of q1 for the top feed (Eq. 6) combined with a simple material balance around 

the top feed tray itself.  The result is, 

Now the slope and intercept for the MOL can be calculated from Eqs. 16-18. 

Eq. 18  is valid because the MOL is connected at one end by x
//
, y

//
 which is point b on 

Figure 2. 

The intersection of the bottom q-line and the MOL is established by equating Eqs. 5 and 

16.  The solution of these equations yields Point d (x
///

, y
///

) on Figure 2.  

Lower operating line.  The lower operating line (LOL) is now very easily determined.  

It is simply the line segment d-e with slope given by, 

Finally, the vapor-liquid flow traffic in the stripping (bottom) section can be conveniently 

computed via Eqs. 21 and 22, 

The program.  The computer program, employing all of the above relationships for 

performing the McCabe-Thiele graphical construction procedure numerically, is listed in 

Table 1.  The program is written in QBASIC and has been adapted to run on a PC 

computer (MS-DOS). 

Lines 40-90 request inputs for the coefficients A, B and C of Eq. 1 and E, F and G for Eq. 

3.  These equations along with Eq. 2 establish the equilibrium y-x curve and T-x 

(isobaric) bubble point curve.  Line 97 provides a command to read in values for the 

distillate, bottoms, upper and lower feed compositions, specification of q1 and q2 for 

upper and lower feeds, and the upper and lower total feed flow rates (mols/hr).   All 

compositions are inputted as mole fraction of the light or lower-boiling component.  In 

Lines 220-230 the values of q1 and q2 are set to 0.999999 if either of them were originally 

inputted as 1.0 (saturated liquid feed).  This specification avoids the occurrence of a 

singularity in either Eq. 4 or 5.  Line 240 requires an inputted value for the ratio of 

operating to minimum reflux ratio ( percentage over minimum), and lines 300 and 310 

represent the componential and overall material balances for computing the distillate and 

bottoms flow rates (mols/hr). 

Lines 340-650 employ all of the equations presented in the text to establish the minimum 

reflux ratio, the slopes and intercepts of the upper (rectifying), middle and lower 

(stripping) operating lines, and the mutual points of intersection of the operating lines and 

the q-lines.       

The remainder of the program is devoted to stage-by-stage computations based on the 

previously established equilibrium curve and operating lines.  Starting with xB, the 

specified bottoms liquid composition exiting the reboiler, the equilibrium vapor 

composition yB exiting the reboiler and the reboiler temperature are both calculated.  

Obviously, this point (xB, yB) lies on the equilibrium curve as shown on Figure 3.  From 

the value of yB the composition of the liquid exiting the bottom theoretical tray of the 

column itself is calculated from the equation of the LOL (Eqs. 19, 20).  Graphically these 

calculated points form the bottom triangle on Figure 3.  This is the first theoretical stage 

in the calculation sequence and represents the reboiler.  In this way the program works its 

way up the column following the LOL and equilibrium curve until the equilibrium vapor 

composition first exceeds y
///

 (Point d).  At this point the program is ready to switch over 

to the MOL.  The number of theoretical stages (triangles) recorded at this point 

correspond to the exact number for the stripping section, including the lower (bottom) 

feed tray and reboiler.  As a result, the bottom feed tray is easily located.  It is basically 

represented by the triangle with one corner on the LOL and the other on the MOL.   

Next the program proceeds to perform stage-by-stage computations up the “middle 

section” by following the MOL.  When the equilibrium vapor composition just exceeds 

y
//
 (Point b), the program then switches over to the UOL (rectifier).  This latter step-off 

(triangle) represents the upper (top) feed tray.  Finally, the program proceeds up the UOL 

and proceeds until the vapor composition becomes equal to or greater than xD, the 

specified overhead product composition.  At this point the program stops and notes the 

last stage computed as the top rectifying tray.  If it is required that the distillate exist as a 

dewpoint vapor, this top theoretical tray represents a partial condenser. 

Example.  Let us consider a fractionator which must be roughly sized to distill two 

separate feed streams into a single bubble point liquid distillate and a single bubble point 

liquid bottoms product.  The fractionation is to be conducted at a mean operating pressure 

of 150 psia and involves a separation of the binary system propylene-1-butene.  The feed 

compositions and desired product purities (specs.) are as follows: 

                   Stream Mole fractions 

     Top Feed Btm. Feed Distillate Bottoms 

 Propylene  0.60  0.30  0.95  0.05 

 1-butene  0.40  0.70  0.05  0.95 

                  1.00  1.00  1.00  1.00 

Feed Conditions: 

    Mols/hr Total Flow  q 

 Top Feed              100   1.0   BP liquid 

 Btm. Feed   100   0.0   DP Vapor 

It is desired to operate this column at a reflux ratio that is 100 % over the minimum 

(R/RM = 2.0).  We are asked to calculate (estimate) the following design parameters: 

1. Tray-by-tray vapor/liquid compositions and temperature profiles 

2. Location of both feed trays 

3. Total number of theoretical trays required 

We can safely assume here that such a nearly ideal and close-boiling system can be 

fractionated under nearly constant molal overflow conditions.  As a result, no heat 

balancing is required in the design calculations, and the McCabe-Thiele method would be 

very appropriate to use.  

At low to moderate pressures the VLE for any mixture of hydrocarbons from the same 

homologous series, such as the olefinic hydrocarbons, closely obeys Raoult’s and 

Dalton’s laws.  For these systems, the vapor-liquid equilibrium ratio is simply equal to 

the ratio of the component vapor pressure to the total system pressure. 

(23)

where xi is the mole fraction of species i in the liquid, and yi is the mole fraction of i in 

the vapor phase.  The isobaric T-x-y coexistence curves for a binary system obeying 

Raoult’s and Dalton’s laws are simply computed from the relations, 

where 1 refers to the lower boiling component and 2 to the higher boiler.  The values of 

K1 and K2 are simply computed from Eq. 23. 

For the system propylene-1-butene at 150 psia, the pure component saturation 

temperatures of propylene and 1-butene are 69 deg. F and 163 deg. F respectively.  At 

intermediate temperatures, the isobaric T-x-y diagram is readily computed from Eqs. 23-

25.  Vapor pressures were read from charts present in the Data Book of the American 

Petroleum Institute (2).  The tables below summarize the results of these calculations: 

   VPi  Psia             Ki = VPi/P  m.f. C3H6 

T, 
o
F P, Psia  C3H6 C4H8  C3H6 C4H8  x y 

69 150  150   -  1.0    -  1.0 1.0 

80 150  176 47  1.173 0.313  0.798 0.937 

100 150  227 63  1.513 0.420  0.530 0.803 

120 150  283 84  1.887 0.560  0.332 0.626 

140 150  353 110  2.353 0.733  0.165 0.387 

150 150  396 125  2.640 0.833  0.092 0.244 

163 150    - 150      - 1.0  0.0 0.0 

Relative volatilities,  = KC3-/K1-C4- : 

     Equilibrium Ratio, Ki 

T, 
o
F  x-C3H6   C3H6  C4H8   

69  1.0   1.0     -     - 

80  0.798   1.173  0.313  3.745 

100  0.530   1.513  0.420  3.603 

120  0.332   1.887  0.560  3.369 

140  0.165   2.353  0.733  3.209 

150  0.092   2.640  0.833  3.168 

163  0.0      -  1.0      - 

The values of  and T above were fitted to Eqs. 1 and 3 respectively using the linear 

regression routine available in the spreadsheet program EXCEL.  The following results 

were obtained: 

2

2

0.3956 1.212849 3.037908 (26)

52.7799 146.474 162.9095 (27)

x x

T x x

    

  
    

Here x is the mole fraction propylene in the liquid, and T is the operating temperature in 

degrees F.  The fits are excellent.  Eq. 26 produces an absolute average deviation of 0.65 

percent for the prediction of , and Eq. 27 yields an absolute average deviation of only 

0.2 percent for the prediction of temperature. 

The detailed output and tray-by-tray calculations generated from the QBASIC program 

are listed in Table 2.  The first portion of the output summarizes all of the input data.  The 

coefficients of Eqs 1 and 3 are printed followed by the product and feed compositions, 

values of q1 and q2, R/Rmin, top and bottom feed flow rates, and the calculated distillate 

and bottoms flow rates based on Eqs. 10 and 11.  Next the minimum and operating reflux 

ratios are printed along with the intersection points (x,y compositions) between the q and 

operating lines. 

The remaining output lists stage-by-stage liquid and vapor mole fractions of propylene in 

1-butene, relative volatilities and temperatures (
o
F).  There are four stripping trays with 

tray 1 being the reboiler and tray no. 4 being the bottom or lower feed tray.  Trays 5 

through 9 are present in the middle section where tray 9 is the top or upper feed tray.  The  

three remaining trays (10 to 12) comprise the rectifier or upper section of the column.  

The vapor composition exiting tray 12 is 0.984 mole fraction propylene which exceeds 

the value for the distillate composition spec. ( 0.95 m.f. propylene).  If a vapor distillate is 

required, then tray 12 represents a partial condenser. 

In summary we have the following theoretical stage profile numbered from the bottom 

up: 

  Tray 

  12     Partial condenser or top tray 

  11 

                        10 

9 Top feed tray 

  8 

  7 

  6           

                 5      

                          4     btm. Feed tray 

                          3 

                          2 

                          1      reboiler  

Finally, the vapor-liquid flow rate traffic in all three column sections are printed. 

If the column employs a total condenser (BP liquid distillate), there will be: 

 1 reboiler (theoretical stage) 

 11 theoretical trays in the column 

     

If the column requires a dew point vapor distillate, the stage breakdown for this particular 

column design would: 

 1 reboiler and 1 (partial) condenser   (2 theoretical stages) 

 10 theoretical trays in the column 

Figure 4 shows the actual McCabe-Thiele graphical construction for this particular 

example.   The computer results agree very nicely with the graphical solution. 

NOMENCLATURE 

A, B and C  Coefficients to Eq. 1 

B1   Bottoms flow rate, mols/hr 

D   Distillate flow rate mols/hr 

E, F and G  Coefficients to Eq. 3 

F1   Top feed flow rate, mols/hr 

F2   Bottom feed flow rate, mols/hr 

I1   Intercept of the rectifying (upper) operating line (UOL) 

I2   Intercept of the middle operating line (MOL) 

I3   Intercept of the lower operating line (LOL or stripper) 

L   Liquid reflux or rectifier liquid flow rate, mols/hr 

L
/
   Liquid flow rate in the middle section, mols/hr 

 L(bar)   Liquid flow rate in the stripping (lower) section, mols/hr 

M1   Slope of the rectifying or upper operating line (UOL) 

M2   Slope of the middle operating line (MOL) 

M3   Slope of the lower operating line (LOL or stripper) 

NOMENCLATURE  Cont.- 

Ntheo or N  Total number of theoretical trays at operating reflux ratio, R 

q1   Moles of liquid introduced to middle section of column resulting 

   from introduction of each mole of top feed 

q2   Moles of liquid introduced to stripping (lower) section of column 

   resulting from introduction of each mole of bottom feed 

R   Operating reflux ratio at some percentage over the minimum, L/D 

RM   Minimum reflux ratio (infinite stages) or (L/D)min 

x   Mole fraction of lower boiling or light component in the liquid 

x
/
   Liquid mole fraction of lower boiling or light component at the 

   intersection of the top q-line and the equilibrium curve 

x
//
   Liquid mole fraction of the lower boiling or light component at the  

   intersection of the top q-line and the upper operating line 

x
///

   liquid mole fraction of the lower boiling or light component at the 

   intersection of the bottom q-line and the middle operating line 

xB   Mole fraction of the lower boiling or light component in the  

   bottoms product 

xD   Mole fraction of the lower boiling or light component in the 

   distillate product 

xF1   Mole fraction of the lower or light component in the top feed 

xF2   Mole fraction of the lower or light component in the bottom feed 

NOMENCLATURE  Cont.- 

y   Mole fraction of the lower boiling or light component in the vapor 

y
/
   Vapor mole fraction of the lower boiling or light component at the 

   intersection of the top q-line and the equilibrium curve 

y
//   

Vapor mole fraction of the lower boiling or light component at the 

   intersection of the top q-line and the upper operating line 

y
///

   Vapor mole fraction of the lower boiling or light component at the 

   intersection of the bottom q-line and the middle operating line 

yB   Mole fraction of the lower boiling or light component in the vapor 

   exiting the reboiler and entering the bottom theoretical tray 

   Relative volatility of the lower boiling component with respect to 

   the higher boiling component or ratio of their equilibrium 

   constants, Klt/Khvy 

 

 

 

 

 

 

 

 



 

                            Table 1 

 

             Listing of the QBASIC Program MCTH2F1.BAS 

 

 

10 LPRINT "2-FEED MCCABE/THIELE BINARY DISTILLATION" 

20 LPRINT 

40 READ A, B, C 

60 LPRINT "A", "B", "C" 

70 LPRINT A, B, C 

80 LPRINT 

85 READ E, F, G 

88 LPRINT "E", "F", "G" 

90 LPRINT E, F, G 

95 LPRINT 

97 READ XD, XB, Z1, Z2, Q1, Q2, F1, F2 

100 LPRINT 

110 LPRINT "DISTILLATE COMPOSITION, XD="; XD 

130 LPRINT "BOTTOMS COMPOSITION, XB="; XB 

150 LPRINT "TOP FEED COMPOSITION, Z1="; Z1 

170 LPRINT "BTM FEED COMPOSITION, Z2="; Z2 

190 LPRINT "TOP FEED THERMAL CONDITION, Q1="; Q1 

210 LPRINT "BTM FEED THERMAL CONDITION, Q2="; Q2 

220 IF Q1 = 1 THEN Q1 = .999999 

230 IF Q2 = 1 THEN Q2 = .999999 

240 INPUT "R/RMIN="; R0 

250 LPRINT "R/RMIN="; R0 

270 LPRINT "TOP FEED RATE, M/H="; F1 

290 LPRINT "BTM FEED RATE, M/H="; F2 

300 D = (Z1 * F1 + Z2 * F2 - XB * (F1 + F2)) / (XD - XB) 

310 B1 = F1 + F2 - D 

320 LPRINT "DISTILLATE RATE, M/H="; D 

330 LPRINT "BTMS RATE, M/H="; B1 

340 X0 = .5 

350 A0 = A * X0 * X0 + B * X0 + C 

360 Y0 = (A0 * X0) / (1 + (A0 - 1) * X0) 

370 X1 = (Y0 - Z1 / (1 - Q1)) * (Q1 - 1) / Q1 

380 E1 = ABS(X1 - X0) 

390 IF E1 < .0001 THEN 420 

400 X0 = X1 

410 GOTO 350 

420 RM = (XD - Y0) / (Y0 - X0) 

430 LPRINT 

440 LPRINT "RMIN="; RM 

450 R = R0 * RM 

460 LPRINT 

470 LPRINT "RACTUAL="; R 

480 M1 = R / (R + 1) 

490 I1 = XD / (R + 1) 

500 X2 = (Z1 / (1 - Q1) - I1) / (M1 - Q1 / (Q1 - 1)) 

510 Y2 = M1 * X2 + I1 

520 LPRINT 



530 LPRINT "INTERSECTION OF UPPER O.L. AND TOP FEED Q-LINE, X,Y="; 

X2, Y2 

540 L = R * D 

550 V = L + D 

560 LPR = L + Q1 * F1 

570 VPR = V - (1 - Q1) * F1 

580 M2 = LPR / VPR 

590 I2 = Y2 - M2 * X2 

600 X3 = (Z2 / (1 - Q2) - I2) / (M2 - Q2 / (Q2 - 1)) 

610 Y3 = M2 * X3 + I2 

620 LPRINT 

630 LPRINT "INTERSECTION OF MIDDLE O.L. AND BTM FEED Q-LINE,X,Y="; 

X3, Y3 

640 M3 = (Y3 - XB) / (X3 - XB) 

650 I3 = XB * (1 - M3) 

660 N = 1 

662 TB = E * XB * XB + F * XB + G 

670 AL = A * XB * XB + B * XB + C 

680 YB = (AL * XB) / (1 + (AL - 1) * XB) 

690 LPRINT 

700 LPRINT "STAGE", "X.LIQUID", "Y,VAPOR", "ALPHA", "TEMP,F" 

710 LPRINT N, XB, YB, AL, TB 

720 Y = YB 

730 X = (Y - I3) / M3 

732 T = E * X * X + F * X + G 

740 AL = A * X * X + B * X + C 

750 Y = (AL * X) / (1 + (AL - 1) * X) 

760 N = N + 1 

770 LPRINT N, X, Y, AL, T 

780 IF Y > Y3 THEN 800 

790 GOTO 730 

800 LPRINT "TOP OF STRIPPING SECTION" 

810 LPRINT 

820 LPRINT 

830 X = (Y - I2) / M2 

832 T = E * X * X + F * X + G 

840 AL = A * X * X + B * X + C 

850 Y = (AL * X) / (1 + (AL - 1) * X) 

860 N = N + 1 

870 LPRINT N, X, Y, AL, T 

880 IF Y > Y2 THEN 900 

890 GOTO 830 

900 LPRINT "TOP OF MIDDLE SECTION" 

910 LPRINT 

920 LPRINT 

930 X = (Y - I1) / M1 

932 T = E * X * X + F * X + G 

935 IF X > XD THEN 990 

940 AL = A * X * X + B * X + C 

950 Y = (AL * X) / (1 + (AL - 1) * X) 

960 N = N + 1 

970 LPRINT N, X, Y, AL, T 

980 GOTO 930 

990 LPRINT "TOP OF RECTIFYING SECTION" 



1000 LPRINT 

1005 LPRINT 

1010 LBAR = LPR + Q2 * F2 

1015 VBAR = VPR - (1 - Q2) * F2 

1020 LPRINT "L AND V RATES, M/H, IN RECTIFUER="; L, V 

1025 LPRINT 

1030 LPRINT "L AND V RATES, M/H, IN MIDDLE SCT="; LPR, VPR 

1035 LPRINT 

1040 LPRINT 

1045 LPRINT "L AND V RATES, M/H, IN STRIPPER="; LBAR, VBAR 

2000 DATA -0.3956,1.212849,3.037908 

2010 DATA 52.7799,-146.474,162.9095 

2020 DATA 0.95,0.05,0.6,0.3,1,0,100,100 

5000 END 



                 Table 2 

 

            Output For Example 

 

 

2-FEED MCCABE/THIELE BINARY DISTILLATION 

 

A   B   C 

-.3956  1.212849  3.037908 

 

E   F   G 

52.7799  -146.474  162.9095 

 

DISTILLATE COMPOSITION, XD = 0.95 

BOTTOMS COMPOSITION, XB = 0.05 

TOP FEED COMPOSITION, XF1 = 0.6 

BTM FEED COMPOSITION, XF2 = 0.3 

TOP FEED THERMAL CONDITION, Q1 = 1 

BTM FEED THERMAL CONDITION, Q2 = 0 

R/RMIN = 2 

TOP FEED RATE, M/H = 100 

BTM FEED RATE, M/H = 100 

 

DISTILLATE RATE, M/H = 88.89 

BTM RATE, M/H = 111.11 

 

RMIN = 0.43094 

RACTUAL = 0.86188 

 

INTERSECTION OF UPPER O.L AND TOP FEED Q-LINE, X,Y = 0.60   0.7880 

 

INTERSECTION OF MIDDLE O.L. AND BTM FEED Q-LINE, X,Y = 0.1427, 0.30 

 

STAGE  X-LIQUID Y-VAPOR ALPHA  Temp., F 

 

  1  0.05  0.14018 3.0976 155.8   REBOILER 

  2  0.0834 0.2221 3.136  151.1 

  3  0.1138 0.2894 3.171  146.9 

  4  0.1388 0.3402 3.199  143.6   BTM FEED TRAY 

TOP OF STRIPPING SECTION 

 

  5  0.1803 0.4165 3.244  138.2 

  6  0.2519 0.5277 3.318  129.4 

  7  0.3561 0.6541 3.420  117.4 

  8  0.4745 0.7609 3.524  105.3 

  9  0.5746 0.8296 3.604  96.2    TOP FEED TRAY  

TOP OF MIDDLE SECTION 

 

 10  0.6899 0.8913 3.686  87.0 

 11  0.8233 0.9461 3.768  78.1 

 12  0.9416 0.9841 3.829  71.8     CONDENSER 

TOP OF RECTIFYING SECTION 

 

 



 

 

Table 2 Continued- 

 

L AND V RATES, M/H, IN RECTIFIER = 76.6    165.5 

 

L AND V RATES, M/H, IN MIDDLE SCT. = 176.6 165.5 

 

L AND V RATES, M/H, IN STRIPPER = 176.6  65.5 

